


Chemical Kinetics: Simple Enzyme Reactions

Gary D. Knott, Ph.D.
Civilized Software, Inc.
12109 Heritage Park Circle
Silver Spring, MD 20906 USA
Tel. (301) 962-3711
Email: csi@civilized.com
URL: <http://www.civilized.com>

Let S be a “substrate” material which is converted by the addition of an enzyme, E , into a molecular complex C which dissociates to yield a product P . We have:



This is, of course, an idealization of a more complex schema, since, in fact, even when S and P have the same molecular weight, energy must be used to, at least momentarily, change and restore the conformation of the enzyme E . But for many situations, our simple model will suffice.

Let $S(t)$, $E(t)$, $C(t)$, and $P(t)$ be the amounts of S , E , C , and P respectively at time t . Then we have:

$$\begin{aligned} dC/dt(t) &= K_1S(t)E(t) - K_2C(t) - K_3C(t), & C(0) &= 0, \\ dP/dt(t) &= K_3C(t), & P(0) &= 0, \\ S(t) &= S_0 - C(t) - P(t), & \text{with } S_0 &= S(0), \\ E(t) &= E_0 - C(t), & \text{with } E_0 &= E(0). \end{aligned}$$

Typically the reaction $C \rightarrow P + E$ is much slower than the complex-formation reaction $S + E \rightleftharpoons C$. Thus after the initial rise of C within time t_0 , we have $dC/dt \approx 0$, so that then

$$\begin{aligned} C(t) &\approx K_1S(t)(E_0 - C(t))/(K_2 + K_3), & \text{or} \\ C(t) &\approx E_0S(t)/[(K_2 + K_3)/K_1 + S(t)], & \text{for } t > t_0. \end{aligned}$$

Thus, from the equation for dP/dt , we have the so-called Michealis-Menten equation:

$$dP/dt(t) \approx K_3E_0S(t)/((K_2 + K_3)/K_1 + S(t)), \quad \text{for } t > t_0.$$

Now the slope $dP/dt(t)$ is measurable by fitting a straight-line to a segment of the kinetic curve for $P(t)$ with $t_0 < t \ll t_e$, where t_e is the time such that $S(t_e) \approx 0$, e.g., when almost all the substrate material has been consumed. Note $S(t) \approx S_0$ for $t_0 < t \ll t_e$ if S_0 is large enough, so one can fit the Michealis-Menten equation to the single data point $(S_0, dP/dt)$ to try to obtain K_3 and $K_m = (K_2 + K_3)/K_1$, the so-called Michealis-Menten constant.

The parameter K_3 may be independently resolved by observing that when S_0 is very large we have $S(t)/(K_m + S(t)) \approx 1$ for $t_0 < t \ll t_e$, so $dP/dt(t) \approx K_3 E_0$. Thus the maximum rate of formation of P for a fixed E_0 value is $K_3 E_0$. Let $K_3 E_0 = V_m$. V_m can be determined by measuring the slope, $dP/dt(t)$, of the linear region of the kinetic curve obtained when S_0 is very large. Now, the Michealis-Menten equation becomes:

$$dP/dt(t)/V_m \approx S(t)/(K_m + S(t)) \quad \text{for } t_0 < t \ll t_e.$$

Note $dP/dt(t)/V_m$ is the relative rate of formation of P , i.e., the proportion of the maximum rate which is achieved. Hence, having measured V_m , one can then measure $dP/dt(t)$ for a relatively-large amount of S , and then obtain K_m from the Michealis-Menten equation as $S_0 V_m / (dP/dt(t)) - S_0$.

The Michealis-Menten constant is the amount of substrate which will yield a product formation rate of $V_m/2$. It thus is the point at which the formation of product becomes increasingly sensitive to a decreasing amount of substrate. The activity of the enzyme for a given amount of substrate is determined directly as $dP/dt(t)$ computed from the Michealis-Menten equation for a given K_m .



The Lineweaver-Burke form of the Michealis-Menten equation is often used because of its linear form. It is:

$$1/(dP/dt(t)) \approx (K_m/V_m)(1/S(t)) + (1/V_m).$$

The Eadie and Dixon form is also often used. It is:

$$S(t)/(dP/dt(t)) \approx S(t)/V_m + K_m/V_m.$$

Actually, as noted above, the reaction $S + E \rightleftharpoons C \rightarrow P + E$ is a fiction. It is commonly used to approximate the situation:




where C is ES -complex, and D is EP -complex and E and F are co-factors. Typically K_4 is negligible, but as the amount of P increases, it may block an appreciable amount of enzyme if K_6 is not nearly zero, and this can render the Michealis-Menten equation useless.

The following is an MLAB tutorial sequence for studying the Michaelis-Menten model relative to the simplified kinetic model.

First define the kinetic model by typing:

```
*FUNCTION C DIFF T(T) = K1*(S0-C-P)*(E0-C)-(K2+K3)*C
*FUNCTION P DIFF T(T) = K3*C
*INITIAL C(0)=0
*INITIAL P(0)=0
*S0 = 10; E0 = 1
*K1 = .2; K2 = .025; K3 = .025
```

Thus we have assumed the true situation is:




starting with 10 moles of substrate and 1 mole of enzyme. We may look at the kinetic behavior of this system over 900 seconds by typing:

```
* Q = INTEGRATE(P DIFF T, C DIFF T, 0:900:5)
* TYPE Q ROW 1:180:10
```

The first column of Q is time, t , the second is $P(t)$, the third is $dP/dt(t)$, the fourth is $C(t)$, and the fifth is $dC/dt(t)$. We can look at the graph of P vs. t by typing:

```
* DRAW Q COL 1:2, LINETYPE dashed
* VIEW
```




Now, let us generate some “laboratory data” about our reaction. We shall use the MLAB normal random number generator to generate normally distributed random numbers. Type:

```
* M = Q COL 1:2 ROW 2:162:5
* E = (NORMRAN ON 0^^NROWS(M))/4
* TYPE E
```

E is a vector of “normal” errors. Now type:

```
* M COL 2 = (M COL 2) + E
* DRAW M, POINTTYPE triangle LINETYPE none
* view
```




M is a matrix of P vs. t points “with error”, as might have been measured in an actual laboratory situation. Now, let us “guess” K_1 , K_2 , and K_3 and try to determine them as functions of M . Type:

```
* K1 = .3; K2 = .01; K3 = .02
* CONSTRAINTS CX={K1>K2, K2>0, K3>0}
* METHOD = GEAR; ERRFAC = .002; MAXITER = 12
* FIT(K1,K2,K3), P to M, CONSTRAINTS CX
```

The control variables `METHOD` and `ERRFAC` are set based on prior experience; this problem is stiff and runs slowly! Our curve-fit “predicts” that K_1 , K_2 , and K_3 are 2.4295, .7975, and .025296 respectively, and resets them accordingly. Note K_1 and K_2 are not even close to .2 and .025, but K_3 is approximately correct. We may observe the graph of this fit by typing:

```
* Q1 = INTEGRATE(P DIFF T, C DIFF T, 0:900:5)
* DRAW Q1 COL 1:2 color green
* VIEW
```




Q_1 is now a matrix of “kinetic concentration and velocity curves” for our reaction as determined by curve-fitting. Let us discard our picture by typing:

```
* DELETE W
```

Now, let us analyze the same reaction using the Michaelis-Menten model. Type:


```
* K1 = .2; K2 = .025; K3 = K2
* FUNCTION MM(S)=VM*S/(KM+S)
* KM = (K2+K3)/K1; VM = K3*E0
* DRAW Q COL (1,3)
* SM = S0-(Q COL 2)-(Q COL 4)
* DRAW (Q COL 1)&'(MM ON SM), LINETYPE dashed
* view
```



Recall Q is the matrix of true curves corresponding to $K_1 = .2$, $K_2 = .025$, and $K_3 = .025$. The matrix SM is computed as the amounts of S at time $t = 0, 5, 10, \dots, 900$. The curves we see are the rate of change, dP/dt vs. t and its Michaelis-Menten approximation.

Now, let us generate two runs of “laboratory” data using the error vector E and obtain the two constants V_m and K_m . Type:

```
* DELETE W
* SO = 500
* Z = INTEGRATE (P DIFF T, C DIFF T, 100:300:10) COL 1:2
* DRAW Z
* Z COL 2 = (Z COL 2)+(NORMRAN ON 0^NROWS(Z))/4
* DRAW Z, LINETYPE none, POINTTYPE crosspt
* FUNCTION Y(T)=A*T+B
* CONSTRAINTS QS = {A > 0, B > 0}
* A = 1; B = 1
* FIT(A,B), Y TO Z, CONSTRAINTS QS
* VM =A
* DRAW POINTS (Y, 50:350!2), LINETYPE dashed
* VIEW
```




We have generated a “straight-line” segment of the P vs. t curve for $S_0 = 500$, drawn it, added some “noise”, shown the simulated points obtained, fit a straight line to these points, set V_m as the slope of this line, and drawn the straight-line fit.

Now, we proceed in the same manner to do another “experiment” to help compute K_m . Type:

```
* DELETE W
* S0 = 10
* Z = INTEGRATE(P DIFF T, C DIFF T,100:300:10) COL 1:2
* Z COL 2 = (Z COL 2)+(NORMRAN ON 0^^NROWS(Z))/4
* FIT(A,B),Y TO Z, CONSTRAINTS QS
* KM = VM*S0/A-S0\
* TYPE VM,KM
```

Now, V_m and K_m are computed. Let us look at the result. Type:

```
* DRAW Q COL (1,3)
* DRAW (Q COL 1)&'(MM ON SM) LINETYPE dotted color red
* DRAW Q1 COL (1,3) LINETYPE DASHED color green
* VIEW
```

Note the dP/dt curve predicted from the kinetic differential equation model is much better than the consistent underestimate predicted by the Michaelis-Menten model.

There is another approach to estimating the Michaelis-Menten constants, V_m and K_m , based on the intersections of various linear plots. This scheme is due to R. Eisenthal and A. Cornish-Bowden (Biochemistry Journal, Vol. 139, pp. 715:730). It is robust and, at the cost of more experiments, allows a confidence region for V_m and K_m to be obtained, without the usual restrictive assumptions. Unfortunately, it often produces poor estimates of V_m and K_m .

Given observations (S_{0i}, H_i) of substrate concentrations and corresponding product-formation velocities (obtained by linear-regression), we can construct lines defined by $V_m/H_i + K_m/S_{0i} = 1$, which may be plotted in K_m, V_m space. The line $\{(K_m, V_m) \mid V_m/H_i + K_m/S_{0i} = 1\}$ is the locus of all (K_m, V_m) pairs which could produce the observation (S_{0i}, H_i) . Each of the (K_m, V_m) points obtained by the intersections of all pairs of these lines is an estimate of the “true” K_m, V_m values. The arithmetic median of the K_m -estimates is the Eisenthal-Cornish-Bowden estimate of K_m , and the arithmetic median of the V_m -estimates is the Eisenthal-Cornish-Bowden estimate of V_m .


We shall simulate ten experiments for $S_0 = 50 : 500 : 50$ and compute the Michaelis-Menten velocity curve based on K_m and V_m as estimated by the Eisenthal-Cornish-Bowden procedure.

```
* FUNCTION VMF(I,J) = (SV[I]-SV[J])/(SV[I]/VV[I]-SV[J]/VV[J])
* FUNCTION KMF(I,J) = (VV[J]-VV[I])/(VV[I]/SV[I]-VV[J]/SV[J])
* FOR I = 1:10 DO \
{SO = 50*I;
```

```

Z = INTEGRATE(P DIFF T, C DIFF T, 100:300:10) COL 1:2;
Z COL 2 = (Z COL 2) + (NORMRAN ON 0^NROWS(Z))/4;
LSQRPT = 8;
FIT(A,B),Y to Z, CONSTRAINTS QS;
SV[I] = S0; VV[I] = A;
};
* D = 1:9^^'9
* D = COMPRESS((LIST(D)&'LIST(D'))*'LIST(D'<=D))
* D COL 1 = (D COL 1) +1
* VM = MEDIAN(VMF ON D)
* KM =MEDIAN(KMF ON D)
* TYPE VM,KM
* DELETE W
* DRAW Q COL (1,3)
* DRAW (Q COL 1)&'(MM ON SM),LINETYPE DASHED
* VIEW

```



Overall the best approach to enzyme kinetics is to try to measure enough points on the kinetic curves of several species, so that direct curve-fitting using the appropriate differential equation model can permit the association and dissociation constants to be found. The Michealis-Menten equation is used only due to the difficulty of obtaining data other than $P(t)$ for $t_0 < t < t_2$. Even then, concurrent use of the kinetic model is useful. An excellent source for mathematical models in enzyme kinetics is: *Enzyme Kinetics* by Kent Plowman, published by McGraw-Hill. Another is *Enzyme Kinetics* by Irwen Segal, published by Wiley.